# SYSTEME LINEARER GLEICHUNGEN

```
> restart;
> with(LinearAlgebra):
> A:=Matrix(4,5,[[1,2,3,4,5],[6,7,8,9,10],[11,12,13,14,15],[16,17,18,19,20]]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 \\ 6 & 7 & 8 & 9 & 10 \\ 11 & 12 & 13 & 14 & 15 \\ 16 & 17 & 18 & 19 & 20 \end{bmatrix}$$

```
> b:=<1,2,3,4>;
```

$$b := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \end{bmatrix}$$

Ab ist die "erweiterte Matrix"

```
> Ab:=<A|b>;
```

$$Ab := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 1 \\ 6 & 7 & 8 & 9 & 10 & 2 \\ 11 & 12 & 13 & 14 & 15 & 3 \\ 16 & 17 & 18 & 19 & 20 & 4 \end{bmatrix}$$

Mit dem Befehl ReducedRowEchelonForm (Ab) wird die Zeilenstufenform von Ab berechnet:

```
> PAb:=ReducedRowEchelonForm(Ab);
```

$$PAb := \begin{bmatrix} 1 & 0 & -1 & -2 & -3 & \frac{-3}{5} \\ 0 & 1 & 2 & 3 & 4 & \frac{4}{5} \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

Mit dem Befehl NullSpace(A) wird eine Basis des Lösungsraums des durch A gegebenen homogenen Systems linearer Gleichungen berechnet:

```
> NullSpace(A);
```

$$\left\{ \begin{bmatrix} 3 \\ -4 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 2 \\ -3 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ -2 \\ 1 \\ 0 \\ 0 \end{bmatrix} \right\}$$

```
> LinearSolve(A,b, free='c');
```

$$\begin{bmatrix} -\frac{3}{5} + c_3 + 2\,c_4 + 3\,c_5 \\ \frac{4}{5} - 2\,c_3 - 3\,c_4 - 4\,c_5 \\ c_3 \\ c_4 \\ c_5 \end{bmatrix}$$

```
> LinearSolve(Ab,free='c');
```

$$\begin{bmatrix} -\frac{3}{5} + c_3 + 2\,c_4 + 3\,c_5 \\ \frac{4}{5} - 2\,c_3 - 3\,c_4 - 4\,c_5 \\ c_3 \\ c_4 \\ c_5 \end{bmatrix}$$

```
> LinearSolve(PAb, free='c');
```

$$\begin{bmatrix} -\frac{3}{5} + c_3 + 2\,c_4 + 3\,c_5 \\ \frac{4}{5} - 2\,c_3 - 3\,c_4 - 4\,c_5 \\ c_3 \\ c_4 \\ c_5 \end{bmatrix}$$

Mit dem Befehl Rank wird der Rang einer Matrix berechnet.

```
> Rank(A);
```

$$2$$

**Elementare Umformungen**

```
> Ab1:=Pivot(Ab,1,1);
```

(Durch elementare Umformungen wird erreicht, dass in der ersten Spalte in allen Zeilen außer der ersten 0 steht).

$$Ab1 := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 1 \\ 0 & -5 & -10 & -15 & -20 & -4 \\ 0 & -10 & -20 & -30 & -40 & -8 \\ 0 & -15 & -30 & -45 & -60 & -12 \end{bmatrix}$$

```
> Ab2:=Pivot(Ab1,2,2);
```

$$Ab2 := \begin{bmatrix} 1 & 0 & -1 & -2 & -3 & \frac{-3}{5} \\ 0 & -5 & -10 & -15 & -20 & -4 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> RowOperation(Ab2,2,-1/5);
```

(Die zweite Zeile wird mit -1/5 multipliziert).

$$\begin{bmatrix} 1 & 0 & -1 & -2 & -3 & \frac{-3}{5} \\ 0 & 1 & 2 & 3 & 4 & \frac{4}{5} \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> M :=
<<1,1,1,4>|<1,1,-2,1>|<3,1,1,8>|<-1,1,-1,-
1>>;
```

$$M := \begin{bmatrix} 1 & 1 & 3 & -1 \\ 1 & 1 & 1 & 1 \\ 1 & -2 & 1 & -1 \\ 4 & 1 & 8 & -1 \end{bmatrix}$$

**Ein System linearer Gleichungen mit genau einer Lösung:**

```
> LinearSolve(M,<0,1,1,0>);
```

$$\begin{bmatrix} \frac{25}{6} \\ \frac{4}{3} \\ \frac{-5}{2} \\ -2 \end{bmatrix}$$

```
> M^(-1);
```

$$\begin{bmatrix} \frac{13}{6} & \frac{5}{2} & \frac{5}{3} & \frac{-4}{3} \\ \frac{4}{3} & 1 & \frac{1}{3} & \frac{-2}{3} \\ \frac{-3}{2} & \frac{-3}{2} & -1 & 1 \\ -2 & -1 & -1 & 1 \end{bmatrix}$$

```
> M^(-1).<0,1,1,0>;
```

$$\begin{bmatrix} \frac{25}{6} \\ \frac{4}{3} \\ \frac{-5}{2} \\ -2 \end{bmatrix}$$

```
> ReducedRowEchelonForm(M);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$